
Algorithms for primary decomposition in Singular

Hans Schönemann

Abstract.

Gröbner bases are the main computational tool available for alge-
braic geometry. Building on top of Gröbner bases algorithms for ideal
theoretical operations (intersection, quotient, saturation, free resolu-
tion,...) will be presented. Combining these algorithms with (multi-
variate) factorization leads to several algorithms for primary decompo-
sition of ideals.

§1. Algebraic Sets and Ideals

1.1. Ideals in Polynomial Rings

Consider the polynomial ring R = K[x1, . . . , xn].
If T ⊂ R is any subset, all linear combinations g1f1 + · · · + grfr,

with g1, . . . .gr ∈ R and fr ∈ T , form an ideal 〈T 〉 of R, called the ideal
generated by T . We also say that T is a set of generators for the
ideal.

Hilbert’s Basis Theorem Every ideal of the polynomial ring
K[x1, . . . , xn] has a finite set of generators.

1.2. Algebraic Sets

The affine n-space over K is the set

An(K) =
{
(a1, . . . , an) | a1, . . . , an ∈ K

}
.

Definition. If T ⊂ R is any set of polynomials, its vanishing
locus in An(K) is the set

V (T ) = {p ∈ An(K) | f(p) = 0∀f ∈ T}.

Every such set is called an affine algebraic set.

The vanishing locus of a subset T ⊂ R coincides with that of the
ideal 〈T 〉 generated by T . So every algebraic set in An(K) is of type V (I)
for some ideal I of R. By Hilbert’s basis theorem, it is the vanishing
locus of a set of finitely many polynomials.

The vanishing locus of a single non-constant polynomial is called a
hypersurface of An(K). According to our definitions, every algebraic
set is the intersection of finitely many hypersurfaces.

Example. The twisted cubic curve in A3(R) is obtained by
intersecting the hypersurfaces V (y − x2) and V (xy − z):

Taking vanishing loci defines a map V which sends sets of polyno-
mials to algebraic sets. We summarize the properties of V :

Proposition.

(i) The map V reverses inclusions: If I ⊂ J are subsets of R, then
V (I) ⊃ V (J).

(ii) Affine space and the empty set are algebraic:

V (0) = An(K). V (1) = ∅.
(iii) The union of finitely many algebraic sets is algebraic:

If I1, . . . , Is are ideals of R, then

s⋃
k=1

V (Ik) = V (
s⋂

k=1

Ik).

(iv) The intersection of any family of algebraic sets is algebraic: If
{Iλ} is a family of ideals of R, then

⋂
λ

V (Iλ) = V

(∑
λ

Iλ

)
.

(v) A single point is algebraic: If a1, . . . , an ∈ K, then

V (x1 − a1, . . . , xn − an) = {(a1, . . . , an)}.

This proposition allows to deal with ideals of polynomials instead of
algebraic sets. The key idea behind Gröbner bases is to reduce problems
concerning arbitrary ideals in polynomial rings to problems concerning
monomial ideals.

§2. Basic definitions for Gröbner bases

2.1. Monomial orderings

The basis ingredient for all Gröbner basis algorithms is the ordering
of the monomials (and the concept of the leading term: the term with
the highest monomial).

A monomial ordering (term ordering) on K[x1, . . . , xn] is a total
ordering < on the set of monomials (power products) {xα|α ∈ Nn}
which is compatible with the natural semigroup structure, i.e. xα < xβ

implies xγxα < xγxβ for any γ ∈ Nn.
An ordering < is called a wellordering iff 1 is the smallest mono-

mial. Most of the algorithms work for general orderings.
Robbiano (cf.[R]) proved that any semigroup ordering can be defined

by a matrix A ∈ GL(n,R) as follows (matrix ordering):
Let a1, . . . , ak be the rows of A, then xα < xβ if and only if there is

an i with ajα = ajβ for j < i and aiα < aiβ. Thus, x
α < xβ if and only

if Aα is smaller than Aβ with respect to the lexicographical ordering of
vectors in Rn.

We call an ordering a degree ordering if it is given by a matrix
with coefficients of the first row either all positive or all negative.

Let K be a field; for g ∈ K[x], g 	= 0, let L(g) be the leading
monomial with respect to the ordering <1 and c(g) the coefficient of
L(g) in g, that is g = c(g)L(g)+ smaller terms with respect to <.

An ordering < is an elimination ordering for xr+1, . . . , xn iff
L(g) ∈ K[x1, . . . , xr] implies g ∈ K[x1, . . . , xr]).

2.2. Examples for monomial orderings

Important orderings for applications are:

• The lexicographical ordering lp, given by the matrix:⎛
⎜⎜⎜⎝

1
1 0

. . .

0 1

⎞
⎟⎟⎟⎠

1we write the terms of a polynomial in decreasing order

resp. ls:

⎛
⎜⎜⎜⎝

−1
−1 0

. . .

0 −1

⎞
⎟⎟⎟⎠

Remark 2.1. The positive lexicographic ordering lp on
K[x1, . . . , xn] is an elimination ordering for x1, . . . , xi

∀1 ≤ i ≤ n.

The definition of rings with these orderings in Singular:
(Each line starting with // is a comment in Singular.)

ring R1=0,(x(1..5)),lp;

ring R2=0,(x(1..5)),ls;

• The weighted degree reverse lexicographical ordering,
given by the matrix

wp :

⎛
⎜⎜⎝

w1 w2 . . . wn

−1
�

0 −1

⎞
⎟⎟⎠

wi > 0∀i, (resp. ws : w1 	= 0, wi ∈ Z ∀i).
If wi = 1 (respectively wi = −1) for all i we obtain the

degree reverse lexicographical ordering, dp (respectively
ds).

The definition of rings with these orderings in Singular:

ring R3=0,(x(1..5)),wp(2,3,4,5,6);

// correspond to w_i:2,3,4,5,6

ring R4=0,(x(1..5)),ws(2,3,4,5,6);

// correspond to w_i:-2,-3,-4,-5,-6

ring R5=0,(x(1..4)),dp;

ring R6=0,(x(1..4)),ds;

• An example for an elimination ordering for xr+1, . . . , xn in
K[x] = Loc<K[x] is given by the matrix

⎛
⎜⎜⎜⎜⎜⎜⎜⎜⎜⎜⎝

0 0 . . . 0 wr+1 wr+2 . . . wn

w1 w2 . . . wr 0 0 . . . 0
−1

�

0 −1
−1

�

0 −1

⎞
⎟⎟⎟⎟⎟⎟⎟⎟⎟⎟⎠

with w1 > 0, . . . , wn > 0. InK[x1, . . . , xr](x1,...,xr)[xr+1, . . . , xn]
= Loc<K[x] it is given by the same matrix with w1 < 0, . . . ,
wr < 0 and wr+1 > 0, . . . , wn > 0.

The definition of a polynomial ring with an elimination
ordering for x3 and x4 in Singular:

ring E=0,(x(1..4)),(a(0,0,1,1),a(1,1),dp);

// correspond to w_i=1 for all i, r=2

// or simpler:

ring EE=0,(x(1..4)),(a(0,0,1,1),dp);

• The product ordering, given by the matrix

⎛
⎜⎜⎜⎝

A1 0
A2

. . .

0 Ak

⎞
⎟⎟⎟⎠

if the Ai define orderings on monomials given by the corre-
sponding subsets of {x1, . . . , xn}. Such an ordering can be
used to compute in

– K(y)[x] (A1 : dp on x, A2 : dp on y)
– (K[y](y))[x] (A1 : dp on x, A2 : ds on y)

– (K[y])[x](x) (A1 : ds on x, A2 : dp on y)
(See [GTZ], [GP], definition 3.1).

The definition of a ring with this ordering in Singular:

ring P=0,(x(1..6)),(dp(4),ds(2));

// correspond to

// a first block of 4 variables with ordering dp

// and a 2nd block of 2 variables with ordering ds

§3. Gröbner bases

Definition 3.1. We define Loc<K[x] := S−1
< K[x] to be the local-

ization of K[x] with respect to the multiplicative closed set S< = {1+ g |
g = 0 or g ∈ K[x]\{0} and 1 > L(g)}.

Remark 3.2. 1) K[x] ⊆ Loc<K[x] ⊆ K[x](x), whereK[x](x)
denotes the localization of K[x] with respect to the maxi-
mal ideal (x1, . . . , xn). In particular, Loc<K[x] is noetherian,
Loc<K[x] is K[x]–flat and K[x](x) is Loc<K[x]–flat.

2) If < is a wellordering then x0 = 1 is the smallest monomial
and Loc<K[x] = K[x] . If 1 > xi for all i, then Loc<K[x] =
K[x](x).

3) If, in general, x1, . . . , xr < 1 and xr+1, . . . , xn > 1 then

1 + (x1, . . . , xr)K[x1, . . . , xr] ⊆ S< ⊆ 1 + (x1, . . . , xr)K[x] =: S,

hence

K[x1, . . . , xr](x1,...,xr)[xr+1, . . . , xn] ⊆ Loc<K[x] ⊆ S−1K[x].

3.1. Definition

Definition 3.3. 1) L(I) denotes the ideal of K[x] generated
by {L(f)|f ∈ I}.

2) f1, . . . , fs ∈ I is called a Gröbner basis of I if
{L(f1), . . . , L(fs)} generates the ideal L(I) ⊂ K[x].

Remark 3.4. The term Gröbner basis is usually only in the case
of well-orderings used, for more general monomial orderings, especially
for local orderings, f1, . . . , fs ∈ I is called a standard basis of I

Singular example: A Gröbner basis computation:

// define a ring R= (Z/32003)[x,y,z]

ring R = 32003, (x,y,z), dp ;

// define 3 polynomials

poly s1 =x^3*y^2 + 151*x^5*y + 169*x^2*y4

+ 151*x^2*y*z3 + 186*x*y^6 + 169*y^9;

poly s2 =x^2*y^2*z^2 + 3*z^8;

poly s3 =5*x^4*y^2 + 4*x*y^5 + 2x^2*y^2*z^3 + y^7 + 11*x^10;

// define the ideal i generated by s1,s2,s3

ideal i = s1, s2, s3;

// compute standard basis j of i

ideal j = std(i);

// display j;

j;

=> j[1]=z8+10668x2y2z2

=> j[2]=y9-567xy6+6250x5y+x2y4+6250x2yz3-5681x3y2

=> j[3]=x10-8728y7-2909x2y2z3-11637x4y2-2909xy5

3.2. Gröbner Bases for Submodules of Free Modules

We consider also module orderings <m on the set of “monomials”
{xαei} of K[x]r =

∑
i=1,...,r K[x]ei which are compatible with the order-

ing < on K[x]. That is for all monomials f, f ′ ∈ K[x]r and p, q ∈ K[x]
we have: f <m f ′ implies pf <m pf ′ and p < q implies pf <m qf .

We now fix an ordering <m on K[x]r compatible with < and denote
it also with <. Again we have the notion of coefficient c(f) and leading
monomial L(f). < has the important property:

L(qf) = L(q)L(f) for q ∈ K[x] and f ∈ K[x]r,
L(f + g) ≤ max(L(f), L(g)) for f, g ∈ K[x]r.

Definition 3.5. 1) L(I) denotes the submodule of K[x]r gen-
erated by {L(f)|f ∈ I}.

2) f1, . . . , fs ∈ I is called a Gröbner basis of I if
{L(f1), . . . , L(fs)} generates the submodule L(I) ⊂ K[x]r.

In Singular submodules of free modules are defined by a set of
generators. These sets are of type module.

3.3. Basic properties of Gröbner Bases

3.3.1. Ideal membership

Definition 3.6. A function NF : K[x]r ×{G|G standardbasis} →
K[x]r, (p,G) 
→ NF (p|G), is called a normal form if for any p ∈ K[x]r

and any G the following holds: if NF (p|G) 	= 0 then L(g) 	 |L(NF (p|G))
for all g ∈ G. NF (g|G) is called the normal form of p with respect
to G.

Lemma 3.7. f ∈ I iff NF (f, std(I)) = 0.

Singular example:

//f defines a trimodal singularity for generic moduli

ring R = 0,(x,y),ds;

int a1,a2,a3=random(1,100),random(-100,1),random(1,100);

poly f = (x^2-y^3)*(y+a1*x)*(y+a2*x)*(y+a3*x);

ideal J = jacob(f);

ideal I = f;

// J:I, ideal of the closure of V(J) \ V(I)

ideal Q = quotient(J,I);

//the Hessian of f

poly Hess = det(jacob(jacob(f)));

//Hess is contained in Q iff NF is 0

reduce(Hess,std(Q));

=> 0

3.3.2. Elimination

Lemma 3.8. Let < be an elimination order for y1, . . . , yn, R =
K[x1, . . . , xr, y1, . . . , yn].
Then std(I) ∩K[x1, . . . , xr] = std(I ∩K[x1, . . . , xr]).

Singular example:

//find the equations from a parametrization t->(t^3,t^4,t^5)

ring R=0,(x,y,z,t),dp;

ideal i=x-t^3,

y-t^4,

z-t^5;

ideal j=eliminate(i,t);

j;

=> j[1]=y2-xz

=> j[2]=x2y-z2

=> j[3]=x3-yz

3.3.3. Geometry of Elimination Elimination means geometrically to
compute the projection π : V (I) ⊂ Kn −→ Kn−k+1.

Definition. Let A ⊂ An(K) and B ⊂ Am(K) be (nonempty) al-
gebraic sets. A map ϕ : A → B is a polynomial map, or a morphism,
if its components are polynomial functions on A. That is, there exist
polynomials f1, . . . , fm ∈ R such that ϕ(p) = (f1(p), . . . , fm(p)) for all
p ∈ A.

The image of a morphism needs not be an algebraic set.
Example. Let π : A2(R) → A1(R), (a, b) 
→ b, be projection of

the xy-plane onto the y-axis. Then π maps the hyperbola C = V (xy−1)
onto the punctured line π(C) = A1(R) \ {0} which is not an algebraic
set.

3.3.4. Hilbert Series

Definition 3.9. Let M be a graded module over K[x]. The Hilbert
series of M is the power series

H(M)(t) =
∞∑

t=−∞
dimKMit

i.

Lemma 3.10. Let < be a (positive or negative) degree ordering
and H(M) the Hilbert function of (the homogenization of) I. Then
H(M) = H(L(M)).

Remark 3.11. It turns out that H(M)(t) can be written in two
useful ways:

(1) H(M)(t) = Q(t)/(1− t)n, where Q(t) is a polynomial in t and
n its the number of variables in K[x].

(2) H(M)(t) = P (t)/(1 − t)dimM where P (t) is a polynomial and
degM = P (1).

(3) vector space dimension dimK(M) = dimK(L(M)).

Remark 3.12. Let < be a degree ordering.

• Krull dimension: dim(M) = dim(L(M)).
• degree (for a positive degree ordering) resp. multiplicity (for a

negative degree ordering) is equal for M and L(M).

Singular example:

// the rational quartic curve J in P^3:

ring R=0,(a,b,c,d),dp;

ideal J=c3-bd2,bc-ad,b3-a2c,ac2-b2d;

// the output of hilb is Q, then P:

hilb(J);

=>// ** J is no standard basis

=>// 1 t^0

=>// -1 t^2

=>// -3 t^3

=>// 4 t^4

=>// -1 t^5

=>

=>// 1 t^0

=>// 2 t^1

=>// 2 t^2

=>// -1 t^3

=>// dimension (proj.) = 1

=>// degree (proj.) = 4

3.3.5. Submodule Membership

Lemma 3.13. (F ) ⊆ (G) iff NF (F, std(G)) = 0.

Singular example:

ring r=0,(x,y,z),dp;

module F=[x,y],[0,y];

module G=[x,0],[0,y],[0,z];

reduce(F,std(G));

=> _[1]=0

=> _[2]=0

3.3.6. Euclidean Algorithm

Lemma 3.14. If < is a wellordering and I = {f, g} ⊆ K[x] then
the computation of the Gröbner basis of I yields the greatest common
divisor of f and g.

Singular example:

ring R=32003,x,dp;

poly f=(x^3+5)^2*(x-2)*(x^2+x+2)^4;

poly g=(x^3+5)*(x^2-3)*(x^2+x+2);

ideal I=f,g;

std(I);

=>_[1]=x5+x4+2x3+5x2+5x+10

// and the expected result:

(x^3+5)*(x^2+x+2);

=>x5+x4+2x3+5x2+5x+10

3.3.7. Gaussian Algorithm

Lemma 3.15. If < is a wellordering and the generators of I are
linear then the computation of the Gröbner basis of I is a Gaussian
algorithm with the columns of matrix(I).

Singular example:

ring R=32003,(x,y,z),dp;

ideal I=22*x+77*y+z-3,

0*x+ 1*y+z-77,

1*x+ 0*y+z+11;

std(I);

=> _[1]=z-58

=> _[2]=y+z-77

=> _[3]=x+z+11



Primary decomposition in Singular 181

3.3.8. Kernel of a Ring Homomorphism

Lemma 3.16. Let Φ be an affine ring homomorphism

Φ : R = K[x1, . . . , xm]/I −→ K[y1, . . . , yn]/(g1, . . . , gs)

given by fi = Φ(xi) ∈ K[y1, . . . , yn]/(g1, . . . , gs) , i = 1, . . . ,m .
Then Ker(Φ) is generated by

(g1(y), . . . , gs(y) , (x1 − f1(y)), . . . , (xm − fm(y))) ∩K[x1, . . . , xm]

in K[x1, . . . , xm]/I .

Remark 3.17. For std(H) ∩R use lemma 3.8.

Singular example:

ring r1=32003,(x,y,z,w),lp;

ring r=32003,(x,y,z),dp;

ideal i=x,y,z;

ideal i1=x,y;

ideal i0=0;

map f=r1,i;

setring r1;

ideal i1=preimage(r,f,i1);

i1;

==> i1[1]=w

==> i1[2]=y

==> i1[3]=x

// the kernel of f

preimage(r,f,i0);

==> _[1]=w

// or, use:

kernel(r,f);

==> _[1]=w

3.3.9. Radical Membership

Lemma 3.18. Let I ⊆ R = Loc<K[x1, . . . , xn], I generated by F .

f ∈ √
I iff 1 ∈ std(F + (yf − 1) ⊆ R[y].

3.3.10. Principal Ideal

Lemma 3.19. I = (F ) is principal (i.e. has a one-element ideal
basis) iff std(F ) has exactly one element.

3.3.11. Trivial Ideal

Lemma 3.20. (F ) is the whole ring R iff std(F ) = {1}.

3.3.12. Module Intersection 1

Lemma 3.21. Let (F ) ⊆ R and (G) ⊆ R.
Then std((F ) ∩ (G)) = std(y(F ) + (1− y)(G)) ∩R in R[y].

Remark 3.22. For std(H) ∩R use lemma 3.8.

Singular example:

ring r1 = 32003,(x,y,z),(c,ds);

poly s1=x2y3+45x6y3+68x4z5+80y6x8;

poly s2=6x5+3y6+8z6;

poly s3=12xyz3+2y3z6;

ideal i1=s1,s2,s3;

ideal i2=s1+s2,s2,s1;

intersect(i1,i2);

=>_[1]=6x5+x2y3+3y6+8z6+45x6y3+68x4z5+80x8y6

=>_[2]=6x5+3y6+8z6

=>_[3]=x2y3+45x6y3+68x4z5+80x8y6

§4. Syzygies

4.1. Definition

Definition 4.1. Let I = {g1, . . . , gq} ⊆ K[x]r.
The module of syzygies syz(I) is ker (K[x]q → K[x]r,

∑
wiei 
→∑

wigi).

Lemma 4.2. The module of syzygies of I is

(g1(x)− er+1, . . . , gq(y)− er+q) ∩ {0}r ×K[x]q

in (K[x1, . . . , xm]/J)q .

Remark 4.3. Use a module ordering with ei > ej∀i ≤ r < j and
the elimination property of lemma 3.8.

Singular example:

ring R=0,(x,y,z),(c,dp);

ideal I=maxideal(1);

// the syzygies of the (x,y,z)

syz(I);

=>_[1]=[0,z,-y]

=>_[2]=[z,0,-x]

=>_[3]=[y,-x]

// syz yields a generating set for the module of syzygies

// but may not be a standard basis !

4.2. Free Resolutions

Iterating the syz command yields a free resolution of a module or
ideal. Singular does this if the res or mres command is used.

Another algorithm due to Schreyer is presented in [S]. It is be used
by the sres command.
Singular example:

ring r=0,(x,y,z),dp;

ideal I=x,y,z;

list Ir=res(I,0);

// print the results:

Ir;

=>[1]:

=> _[1]=z

=> _[2]=y

=> _[3]=x

=>[2]:

=> _[1]=-y*gen(1)+z*gen(2)

=> _[2]=-x*gen(1)+z*gen(3)

=> _[3]=-x*gen(2)+y*gen(3)

=>[3]:

=> _[1]=x*gen(1)-y*gen(2)+z*gen(3)

4.3. Kernel of a Module Homomorphism

Definition 4.4. Let R = K[x1, . . . , xn]/(h1, . . . , hp) ,
A ∈ Mat(m× r,R) and B ∈ Mat(m× s,R) then define

modulo(A,B) := ker(Rr A−→ Rm/Im(B))

(modulo(A,B) is the preimage of B under the homomorphism given by
A.)

Lemma 4.5. Let { (αi, βi
, γ

i
) | i = 1, . . . , k } ⊂ Rr+s+p =: RN be

a generating set of syz(D) where

C =

⎛
⎜⎜⎜⎝

h1 · · · hp 0 · · · · · · · · · · · · · · ·
0 · · · 0 h1 · · · hp 0 · · · · · ·
...

...
0 · · · · · · · · · · · · · · · h1 · · · hp

⎞
⎟⎟⎟⎠

∈ Mat(m× pm,R)

and

D =

⎛
⎜⎝

a11 · · · a1r b11 · · · b1s c11 · · · c1,pm
...

...
...

...
...

...
am1 · · · amr bm1 · · · bms cm1 · · · cm,pm

⎞
⎟⎠

∈ Mat(m× r + s+ pm,R)

Then

modulo (A,B) := (α1 . . . αk ) ∈ Mat(r × k,R)

(see Lemma 4.2.)

Remark 4.6. In practice, one need not compute the entire syzygy
module of D: it is better to find modulo(A,B) as:⎛
⎜⎜⎜⎜⎜⎜⎜⎝

a11 · · · a1r b11 · · · b1s c11 · · · c1,pm
...

...
...

...
...

...
am1 · · · amr bm1 · · · bms cm1 · · · cm,pm

1 0 0 · · · · · · · · · · · · 0
0 . . . 0 0 · · · · · · · · · · · · 0
0 1 0 · · · · · · · · · · · · 0

⎞
⎟⎟⎟⎟⎟⎟⎟⎠

∩

⎛
⎜⎜⎜⎜⎜⎜⎜⎜⎝

0
...
0
R
...
R

⎞
⎟⎟⎟⎟⎟⎟⎟⎟⎠

(see Sections 4.2, 3.8.)

4.4. Module intersection 2

Let R be an affine ring, and let I, J,K ⊆ R be ideals. One can
compute generators for the intersection L = I ∩ J ∩K in the following
way: L is the kernel of the R-module homomorphism φ : R → R/I ⊕
R/J ⊕R/K which sends 1 to (1,1,1).

Lemma 4.7.

I ∩ J ∩K = modulo(

⎛
⎝ 1

1
1

⎞
⎠ ,

⎛
⎝ I 0 0

0 J 0
0 0 K

⎞
⎠).

4.5. Ideal Quotient

Lemma 4.8. The quotient (I : J) of two ideals I = (a1, . . . , ar)
and J = (b1, . . . , bs) in R is the kernel of the map

R −→ R/I ⊕ . . .⊕R/I
1 
−→ (b1, . . . , bs)

It can be computed as

(I : J) = modulo
(
(b1| . . . |bs)T , (a1| . . . |ar)⊕ . . .⊕ (a1| . . . |ar)

)

4.6. Saturation

The saturation (I : J)∞ of I with respect J can be computed by
computing (I : J), ((I : J) : J), . . . until it stabilizes.

4.7. Annihilator of a Module

Lemma 4.9. Let R = Loc<K[x1, . . . , xn]/(h1, . . . , hp) , M ⊆ Rm.
AnnR(R

m/M) := { g ∈ R | gRm ⊂ M } is generated by first entries of
syzygies of the module⎛

⎜⎜⎜⎜⎝
e1 M 0 · · · 0

e2 0 M
. . .

...
...

...
. . .

. . . 0
em 0 · · · 0 M

⎞
⎟⎟⎟⎟⎠

where ei is the i-th unit vector in Rm.
(We identify a matrix with the module generated by it columns.)

§5. Primary Decomposition

5.1. Motivation: Decomposition of Algebraic Sets

Example

ring R=0,(x,y,z),dp;

ideal K=yz,xz;

LIB "primdec.lib";

primdecGTZ(K);

=>_[1][1]=z

=>_[1][2]=z

=>_[2][1]=x,y

=>_[2][2]=x,y

5.2. Definition.

A proper ideal Q of a ring R is said to be primary if f, g ∈ R,
fg ∈ Q and f 	∈ Q implies g ∈ √

Q. In this case, P =
√
Q is a prime

ideal, and Q is also said to be a P -primary ideal. Given any ideal
I of R, a primary decomposition of I is an expression of I as an
intersection of finitely many primary ideals.

Now suppose that R is Noetherian. Then every proper ideal I of R
has a primary decomposition. We can always achieve that such a decom-
position I =

⋂r
i=1 Qi is minimal. That is, the prime ideals Pi =

√
Qi

are all distinct and none of the Qi can be left out. In this case, the Pi are
uniquely determined by I and are referred to as the associated primes

of I. If Pi is minimal among P1, . . . , Pr with respect to inclusion, it is
called a minimal associated prime of I.

The minimal associated primes of I are precisely the minimal prime
ideals containing I. Their intersection is equal to

√
I. Every primary

ideal occurring in a minimal primary decomposition of I is called a
primary component of I. The component is said to be isolated if
its radical is a minimal associated prime of I. Otherwise, it is said to
be embedded. The isolated components are uniquely determined by I,
the others are far from being unique.

From the definitions, it is clear that there is a number of different
tasks coming with primary decomposition. These range from comput-
ing radicals via computing the minimal associated primes to computing
a full primary decomposition. A variety of corresponding algorithms
is implemented in the Singular library primdec.lib. The two main
algorithms for computing a full primary decomposition are primdecSY

and primdecGTZ. A detailed description is given in [DGP].

5.3. Computational Primary Decomposition: History of
Algorithms

First approaches for algorithms to compute the primary decomposi-
tion of polynomial ideals were given by Grete Herrmann (1926): generic
projections, computed via resultants, reduce the task to the case of hy-
persurfaces (i.e. the case of principal ideals). The case of principal ideals
was handled by factoring polynomials.

A more modern algorithm was proposed by Wu in Maple (Wu-Ritt-
process, 1989): compute the set of all minimal associated primes via
characteristic sets, and use Gröbner base techniques for saturation steps
(see ”Algorithm of Shimoyama-Yokoyama” (SY) below).

The algorithm by Gianni,Trager,Zacharias in AXIOM (1988) via
factorization is the basis of primdecGTZ in Singular, which reduces to
the zerodimensional case (via Gröbner bases for projection) and handles
this case by polynomial factorization.

The algorithm by Eisenbud, Huneke, Vasconcelos (1992) avoids the
time consuming elimination and decomposes into equidimensional parts.

Singular implements GTZ, SY (1998), EHV (2001) in the libraries
primdec.lib and ehv.lib.

5.4. Computational Primary Decomposition by the Algo-
rithm of Gianni, Trager, Zacharias

The starting point of the GTZ-algorithm is the following simple
observation:

Lemma (Splitting Tool). If I ⊂ R is an ideal, if h ∈ R is a
polynomial, and if m ≥ 1 is an integer such that I : 〈h〉∞ = I : 〈h〉m,
then

I =
(
I : 〈h〉m) ∩ 〈I, hm〉 .

The key result on which the algorithm is based specifies which poly-
nomials h are considered:
If I : f = I : f2 for some f , then I = (I : f) ∩ (I, f).
If fg ∈ I and (f, g) = R, then I = (I, f) ∩ (I, g).

If fg ∈ I, then
√
I =

√
(I, f) ∩√

(I, g).

If fr ∈ I, then
√
I =

√
(I, f).

If J ⊆ R an ideal, then
√
I =

√
I : J ∩√

I + J =
√
I : J ∩√

I : (I : J).
5.4.1. Primary Decomposition: Reduction to Dimension 0
Proposition Let I � K[x] = R be a proper ideal, and let u ⊂ x

be a subset of maximal cardinality such that I ∩K[u] = {0}. Then:
• The ideal I K(u)[x\u] ⊂ K(u)[x\u] is zero-dimensional.
• Let > = (>x\u , >u) be a global product ordering on K[x],
and let G be a Gröbner basis for I with respect to >. Then G
is a Gröbner basis for I K(u)[x\u] with respect to the mono-
mial ordering obtained by restricting > to the monomials in
K[x\u]. Further, if h ∈ K[u] is the least common multiple
of the leading coefficients of the elements of G (regarded as
polynomials in K(u)[x\u]), then

I K(u)[x\u] ∩K[x] = I : 〈h〉∞ .

• All primary components of the ideal I K(u)[x\u] ∩K[x] have
the same dimension, namely dim I.
Further, if I K(u)[x\u] = Q1 ∩ . . . ∩Qr is the minimal pri-
mary decomposition, then

I K(u)[x\u] ∩K[x] = (Q1 ∩K[x]) ∩ . . . ∩ (Qr ∩K[x])

is the minimal primary decomposition, too.

If > is a global monomial ordering on K[x], then every subset u ⊂ x
of maximal cardinality satisfying L>(I) ∩K[u] = {0} is also a subset of
maximal cardinality such that I ∩K[u] = {0}.

By recursion, the proposition allows us to reduce the general case of
primary decomposition to the zero-dimensional case. In turn, if I ⊂ K[x]
is a zero-dimensional ideal “in general position” (with respect to the le-
xicographic order satisfying x1 > · · · > xn), and if hn is a generator
for I ∩ K[xn], the minimal primary decomposition of I is obtained by

factorizing hn. In characteristic zero, the condition that I is in gen-
eral position can be achieved by means of a generic linear coordinate
transformation

5.4.2. Zero-dimensional Primary Decomposition The lexicographi-
cal Gröbner basis of a zero-dimensional ideal I contains one polynomial
f of only the last variable. Let fα1

1 ....fαr
r = f the decomposition of f in

irreducible factors.
Then the minimal primary decomposition of I is given by

I = ∩r
k=1(I, f

αk

k )

5.5. Computational Primary Decomposition by the Algo-
rithm of Eisenbud, Huneke, Vasconcelos

The algorithm by Eisenbud, Huneke, Vasconcelos avoids the time
consuming elimination and decomposes into equidimensional parts Its
main splitting tools is (cf. [EHV])

Proposition. If I ⊆ R = K[x1, ..xn] is an ideal, then the equidi-

mensional hull of I E(I) is AnnExtn−d
R (R/I,R), where d = dim(I).

5.6. Computational Primary Decomposition by Charac-
teristic Series

5.6.1. Characteristic Series Let < be the lexicographical ordering
on R = K[x1, ..., xn] with x1 < ... < xn. For f ∈ R let lvar(f)
(the leading variable of f) be the largest variable in f , i.e., if f =
as(x1, ..., xk−1)x

s
k + ... + a0(x1, ..., xk−1) for some k ≤ n then lvar(f) =

xk.
Moreover, let ini(f) := as(x1, ..., xk−1). The pseudo remainder r =

prem(g, f) of g with respect to f is defined by the equality ini(f)a · g =
qf + r with deglvar(f)(r) < deglvar(f)(f) and a minimal. A set T =

{f1, ..., fr} ⊂ R is called triangular if lvar(f1) < ... < lvar(fr). Moreover,
let U ⊂ T , then (T,U) is called a triangular system, if T is a triangular
set such that ini(T ) does not vanish on V (T ) \ V (U)(=: V (T \ U)).

T is called irreducible if for every i there are no di,f
′
i ,f

′′
i such that

lvar(di) < lvar(fi) = lvar(f ′
i) = lvar(f ′′

i ),

0 	∈ prem({di, ini(f ′
i), ini(f

′′
i )}, {f1, ..., fi−1}),

prem(difi − f ′
if

′′
i , {f1, ..., fi−1}) = 0.

Furthermore, (T,U) is called irreducible if T is irreducible.
The main result on triangular sets is the following:

Let G = {g1, ..., gs} ⊂ R, then there are irreducible triangular sets

T1, ..., Tl such that V (G) =
⋃l

i=1(V (Ti\Ii)) where Ii = {ini(f) | f ∈ Ti}.

Such a set {T1, ..., Tl} is called an irreducible characteristic series
of the ideal (G).
Singular example:

ring R= 0,(x,y,z,u),dp;

ideal i=-3zu+y2-2x+2,

-3x2u-4yz-6xz+2y2+3xy,

-3z2u-xu+y2z+y;

print(char_series(i));

=>_[1,1],3x2z-y2+2yz,3x2u-3xy-2y2+2yu,

=>x, -y+2z, -2y2+3yu-4

char_series(i);

=>_[1,1]=-9x4y2+18x5-12x2y3-18x4+24x3y+9xy3+2y4

-24x2y+8xy2-8y2

=>_[1,2]=3x2z-y2+2yz

=>_[1,3]=3x2u-3xy-2y2+2yu

=>_[2,1]=x

=>_[2,2]=-y+2z

=>_[2,3]=-2y2+3yu-4

5.6.2. Algorithm of Shimoyama-Yokoyama The SY algorithm (cf
[SY])

• computes the set of all minimal associated primes {P1, ..., Pl}
of the ideal I,

• computes pseudo primary components Q′
i and a remaining

ideal (fe1
1 , ..., fel

l ) with
√
Q′

i = Pi and I = Q′
1 ∩ ... ∩ Q′

l ∩
(I + (fe1

1 , ..., fel
l )),

• decomposes all pseudo primary components Q′
i into a primary

component Qi and a remaining ideal I ′i, s.t. Q
′
i = Qi ∩ I ′i,

• decomposes the remaining ideals I ′i, I + (fe1
1 , ..., fel

l ) by apply-
ing SY recursively.

5.7. Preprocessing: Factorizing Buchberger Algorithm

The factorizing Buchberger algorithm is the combination of
Buchberger algorithm with factorization: each new element for the
Gröbner basis will be factorized, and, if reducible, used to split the com-
putation into several branches corresponding to the factors. Applied to
an ideal I = (f1, ..., fs) it computes a list of Gröbner bases G1, ..., Gr

such that
V (I) = V (G1) ∪ ... ∪ V (Gr).

The V (Gi) need not be irreducible, so this algorithm is mainly used
as a preprocessing step. See [C].

